Maple Lecture 19. Integration and Summation

Integration is one of the highlights of computer algebra, as education is a "killer application" of symbolic computation. Integrals occur everywhere in science and engineering. This lecture matches [1, Chapter 10].

19.1 Indefinite Integration

Just as diff/Diff, we have the int/Int commands.

```
[> integrand := (x^2 - 1)/(x^5+1);
[> exint := Int(integrand,x);
[> value(exint) = int(integrand,x);
[> printlevel := 30;                  # see what goes on
[> valint1 := int(integrand,x);
```

We just discovered that int has a remember table.

```
[> printlevel := 0:                   # reset value of printlevel
[> forget(int);                       # clear remember table of int
[> printlevel := 30:
[> valint2 := int(integrand,x);
[> printlevel := 0:
```

Let us check the answer:

```
[> integrand1 := diff(valint1,x);
[> normint := normal(integrand1); integrand;
```

It is far from obvious that both expressions are equivalent. Let us do a numerical check...

```
[> fnormint := unapply(normint,x): fintegrand := unapply(integrand,x):
[> evalf(fnormint(.1212)); fintegrand(.1212);
```

19.2 Definite Integration

We use antiderivatives to compute definite integrals, applying the fundamental theorem of calculus.

Suppose we wish to compute $\int_a^b \frac{1}{x^2} dx$.

```
[> ad := int(1/x^2,x);                # first compute antiderivative
[> funad := unapply(ad,x);            # ready for evaluation
[> valint := funad(2) - funad(1);
```

We can do this all at once:

```
[> int(1/x^2,x=1..2);
```

Assume we wish to compute the integral of $1/x^2$ over $[-1,+1]$:

```
[> funad(1) - funad(-1);
```

We could also have done it purely symbolically, leaving the end points as parameters.

```
[> symbolic_integral := subs(x=b,ad) - subs(x=a,ad);
[> subs(a=-1,b=1,symbolic_integral);
```

However:

```
[> numerical_integral := int(f,x=-1..+1);
```

The numerical value is right, watch the picture:

```
[> plot(1/x^2,x=-1..1,view=[-1..1,0..100]);
```

The experiment above shows that we have to be cautious when using formal results.

19.3 Numerical Integration

Not every function has a symbolic antiderivative.

```
[> integrand := exp(cos(x));
[> valint := int(integrand,x=0..Pi);
```

Maple returns the integral unevaluated. But we can find a numerical approximation:

```
[> evalf(valint,20);
```

For convenience, we may want to introduce a macro "numint" that calls the evalf after the Int:

```
[> macro(numint=evalf@Int):          # observe the Int, not the int...
[> numint(integrand,x=0..Pi,20);
```

Notice that `eval@Int` would first symbolically compute the integral before applying the numerical evaluation command. This is not what we want: we want Maple to integrate directly using numerical methods. The use of the "inert" version of a command merits some attention. Previously, we found the inert useful for pretty printing, but here we see it can be used to define a function.

19.4 Integral Transforms

Integral transforms turn differential equations into algebraic equations.
Suppose we wish to solve $y'' + y = \sin(2t)$, with initial conditions: $y(0) = 2, y'(0) = 1$.

```
[> diffeq := diff(y(t),t$2) + y(t) = sin(2*t);
[> inits := y(0) = 2, D(y)(0) = 1;
```

Here is how we do everything in one command:

```
[> dsolve({diffeq,inits},y(t),method=laplace);
```

If you had to do this "by hand", you could work as follows:

```
[> with(inttrans):                   # Laplace is just one of the transforms
[> alias(Y(s)=laplace(y(t),t,s));
[> lp := laplace(diffeq,t,s);
[> slp := subs(inits,lp);            # use initial conditions
[> slp_sol := Y(s)=solve(slp,Y(s));
[> sol := invlaplace(slp_sol,s,t);
```

Like the Fourier transforms, with the Laplace transform one goes from the time to the frequency domain.

19.5 Assisting Maple's Integrator

Sometimes we have to make extra assumptions to arrive at a simpler value for the integral. Older versions of Maple (e.g., Maple 7) would simplify this integral:

```
[> int(1/x^2,x=a..b);
```

The current versions of Maple refuse to give a symbolic formula for this integral. We may simplify with additional assumptions on the limits:

```
[> int(1/x^2,x=a..b) assuming a > 0, b > 0;  # local assumptions
```

or, imposing global assumptions on the limits:

```
[> assume(a>0,b>0):
[> int(1/x^2,x=a..b);
```

19.6 Summation

In some cases, Maple finds explicit formulas for the value of a sum.

```
[> s := sum('i','i'=1..n);
```

There is also the inert version, the Sum command:

```
[> sinf := Sum('1/i^2','i'=1..infinity);
[> value(sinf) = evalf(sinf,20);
```

19.7 Assignments

1. Compute $\int x^n e^x dx$ for a general integer n. Check the result for some randomly chosen values for n.
2. Let F be the function defined by $F(T) := \int_1^T \frac{\exp(-t^2 T)}{t} dt$.
   - (a) Define the corresponding Maple function F to return numerical approximations for $F(T)$. Use it to compute $F(2)$. Compare the value for $F(2)$ to $\int_1^2 \frac{\exp(-2t^2)}{t} dt$.
   - (b) Compute the derivative funtion of F. What is $F'(2)$? Compare the value $F'(2)$ to $\frac{F(2+h)-F(2)}{h}$ for sufficiently small values of h.
3. Compute $\int_0^\infty \frac{\ln(x)}{(x+a)(x-1)} dx$ for positive a.
4. Show that $\sum_{k=1}^{n} k^3 = \frac{1}{4} n^2 (n+1)^2$.
5. Compute the sum $\sum_{k=1}^{\infty} \frac{k^2+k-1}{(k+2)!}$.
6. Consider $\frac{x-y}{(x+y)^3}$ for x and y each ranging between 0 and 1.
   - (a) Compute $\int_0^1 \left(\int_0^1 \frac{x-y}{(x+y)^3} dy \right) dx$.
   - (b) Compute $\int_0^1 \left(\int_0^1 \frac{x-y}{(x+y)^3} dx \right) dy$.

   You should notice that Maple returns two different values.
   Explain why this happened (make a plot). Is there a correct value for the integral?
   Give the Maple commands with results to motivate and illustrate your explanations.
7. Define the function $f(k) = \int_0^1 x \sin(2\pi k x) dx$. What is $f(1)$?
   What is $f'(1)$, its derivative at 1?
   Using the Maple definition of f (*do NOT recompute the integral!*),
   how can you see the symbolic formula Maple uses to evaluate f?

References

[1] A. Heck. *Introduction to Maple*. Springer-Verlag, third edition, 2003.